```
> S1 := [x^2+y^2=1,x+y=0];
```

$$S1 := [x^2 + y^2 = 1, x + y = 0]$$

```
> with(plots):
> implicitplot( S1, x=-2..2, y=-2..2, grid=[100,100] );
```




```
> solve( S1, {x,y} );
```

$$\{x = -RootOf(2\_Z^2 - 1), y = RootOf(2\_Z^2 - 1)\}$$

```
> _EnvExplicit := true;
  solve( S1, {x,y} );
```

$$\_EnvExplicit := true$$

$$\left\{x = -\frac{1}{2}\sqrt{2}, y = \frac{1}{2}\sqrt{2}\right\}, \left\{x = \frac{1}{2}\sqrt{2}, y = -\frac{1}{2}\sqrt{2}\right\}$$

```
> S2 := [x+y-z=0, x+z=1];
```

$$S2 := [x + y - z = 0, x + z = 1]$$

```
> solve(S2,{x,y,z});
```

$$\{x = 1 - z, y = -1 + 2z, z = z\}$$

```
> plots[implicitplot3d]( S2, x=-2..2, y=-2..2, z=-2..2, color=[red,
  blue], style=patchnogrid );
```




```
> S3 := [z=x*y,y=x^2];
```

$$S3 := [z = x\,y,\ y = x^2]$$

```
> P1 := implicitplot3d( S3, x=-2..2, y=-2..2, z=-2..2, color=[red,blue],
  style=patchcontour );
```

$P1 :=$




```
> solve( S3, {x,y,z} );
```

$$\{x = x,\ y = x^2,\ z = x^3\}$$

```
> P2 := spacecurve( [t,t^2,t^3], t=-1.5..1.5, color=green, thickness=10
```

```
);
```

$P2 :=$




```
> display([P1,P2]);
```




```
> S1;
```

$$[x^2 + y^2 = 1, x + y = 0]$$

```
> S3 := [x^2+y^2+z^2=4, x^2-y^2-z^2=0, x*y-z^2=1];
```

$$S3 := [x^2 + y^2 + z^2 = 4, x^2 - y^2 - z^2 = 0, xy - z^2 = 1]$$

```
> implicitplot3d( S3, x=-3..3, y=-3..3, z=-3..3, color=[red,green,blue],
```

```
style=patchcontour );
```




```
> G1 := Groebner[Basis]( S1, plex(x,y) );
```

Error, (in Groebner:-Basis) the first argument must be a list or set of polynomials or a PolynomialIdeal

```
> S1 := [x^2+y^2-1, x+y];
```

$$S1 := [x^2 + y^2 - 1, x + y]$$

```
> S2 := [x+y-z, x+z-1];
```

$$S2 := [x + y - z, x + z - 1]$$

```
> S3 := [x^2+y^2+z^2-4, x^2-y^2-z^2, x*y-z^2-1];
```

$$S3 := [x^2 + y^2 + z^2 - 4, x^2 - y^2 - z^2, x\,y - z^2 - 1]$$

```
> G1 := Groebner[Basis]( S1, plex(x,y) );
```

$$G1 := [2\,y^2 - 1, x + y]$$

```
> G2 := Groebner[Basis]( S2, plex(x,y,z) );
```

$$G2 := [1 - 2\,z + y, x + z - 1]$$

```
> G3 := Groebner[Basis]( S3, plex(x,y,z) );
```

$$G3 := [z^4 + 4\,z^2 - 3, y^2 + z^2 - 2, -y\,z^2 + 3\,x - 3\,y]$$

```
> factor(G3[1]);
```

$$z^4 + 4\,z^2 - 3$$

```
> sols := solve( G3[1] = 0, z );
```

$$sols := I\sqrt{2+\sqrt{7}},\ -I\sqrt{2+\sqrt{7}},\ \sqrt{-2+\sqrt{7}},\ -\sqrt{-2+\sqrt{7}}$$

```
> evalf(sols);
```

$$2.155400499\,I,\ -2.155400499\,I,\ 0.8035865299,\ -0.8035865299$$

```
> G := subs( z=sqrt(-2+sqrt(7)), G3 );
```

$$G := \left[\left(-2+\sqrt{7}\right)^2 - 11 + 4\sqrt{7},\ y^2 - 4 + \sqrt{7},\ -y\left(-2+\sqrt{7}\right) + 3\,x - 3\,y\right]$$

```
> G := simplify(G);
```

$$G := \left[0,\ y^2 - 4 + \sqrt{7},\ -y\sqrt{7} + 3\,x - y\right]$$